

# Paris Conférence *Mathematica* 2010

## Garantir le calcul numérique malgré les incertitudes

Calculer avec des intervalles

*Yves.Papegay@sophia.inria.fr*

*INRIA Sophia Antipolis Méditerranée*

---

### Quelques considérations arithmétiques

Habités que nous sommes à calculer dans des arithmétiques exactes - symboliques, entières ou rationnelles - nous avons tendance à oublier que les ordinateurs et autres calculateurs électroniques, malgré leur omnipotence et omniscience affichée ne savent en général représenter - et encore moins manipuler - qu'une infime partie des nombres que l'on dit réels malgré leur abstraction : les **nombres flottants**.

#### Nombres flottants

Les nombres flottants sont caractérisés par deux entiers qui définissent la **base** et la **précision** de leur représentation.

Ainsi, en base 10 et avec une précision de 6,  $\pi$  est représenté par 3.14159 ... et avec une précision de 80 par

```
N[ $\pi$ , 80]
```

```
3.1415926535897932384626433832795028841971693993751058209749445923078164062862090
```

La plupart des machines (dites 64 bits) représentent les nombres en base 2 - avec 52 bits pour la mantisse, 1 bit pour le signe et 11 bits pour l'exposant - ce qui a pour conséquence directe que seule une partie des nombres décimaux est représentable.

#### exemple de 0.1

Par exemple, en base 2, le nombre décimal 0.1 est représenté par

```
BaseForm[0.116, 2]
```

```
0.00011001100110011001100110011001100110011001100110011001100110102
```

or, si l'on calcule exactement ce que cette représentation vaut exactement

```

ToNumber[RealDigits[0.0001100110011001100110011001100110011001100110011010], 2]
N[% , 60]
  3 602 879 701 896 397
  36 028 797 018 963 968
0.100000000000000005551115123125782702118158340454101562500000

```

ce qui est légèrement supérieur à 0.1, et la représentation immédiatement précédente correspond à un nombre légèrement inférieur à 0.1

```

N[ToNumber[RealDigits[0.0001100110011001100110011001100110011001100110011001], 2], 60]
0.0999999999999999916733273153113259468227624893188476562500000

```

### plus petit nombres

Cette représentation est donc par essence discrète. Sa **granularité** dépend du nombre de bits utilisés pour représenter un nombre, mais aussi de la grandeur de ce nombre.

Elle est habituellement définie au voisinage de 1 par le *plus petit* nombre qui ajouté à 1 produit un résultat strictement supérieur à 1.

```

$MachineEpsilon
2 ^ -52 // N
2.22045 × 10-16
2.22045 × 10-16
x = 1. + $MachineEpsilon
y = 1. + ($MachineEpsilon/2)
1.
1.
x - 1.
y - 1.
2.22045 × 10-16
0.

```

Cette granularité n'est pas absolue, mais relative (proportionnelle) au nombre au voisinage duquel on la calcule.





## instabilité numérique

L'instabilité numérique de certains calculs est parfois intrinsèque aux expressions que l'on évalue, parfois relié à la manière dont on conduit cette évaluation, mais rarement prévisible simplement ou évidente. En voici un exemple qui illustre bien la nécessité de certifier les résultats des calculs.

Le polynôme ci-dessous est dû à Rump.

```
RumpFunc[x_, y_] := (1335 / 4 - x^2) y^6 + x^2 (11 x^2 y^2 - 121 y^4 - 2) + (11 / 2) y^8 + x / (2 y)
RumpFunc[77 617, 33 096]
% // N
54 767
-----
66 192
-0.827396
```

Ce premier calcul est exacte puisqu'il est réalisé avec des nombre entiers et rationnels. Voici le même calcul avec des nombres flottants.

```
RumpFunc[77 617., 33 096.]
0.
```

Plus troublant encore est le résultat obtenu si l'on remplace 11/2 par 5.5 dans la définition du polynôme.

```
RumpFuncN[x_, y_] := (1335 / 4 - x^2) y^6 + x^2 (11 x^2 y^2 - 121 y^4 - 2) + (5.5) y^8 + x / (2 y)
RumpFuncN[77 617, 33 096]
1.18059 × 1021
```

Le recours à une arithmétique d'intervalles permet d'obtenir un résultat juste et de comprendre le phénomène ... même si il n'est pas très intéressant d'un point de vue calculatoire.

```
RumpFunc[Interval[77 617.], Interval[33 096.]]
RumpFuncN[Interval[77 617.], Interval[33 096.]]
Interval[{-3.89595 × 1022, 3.65983 × 1022}]
Interval[{-3.89595 × 1022, 3.65983 × 1022}]
```

---

## Arithmétique par intervalles et évaluation des expressions

### Propriétés de l'arithmétique

En étant étendues aux intervalles, la soustraction et la division perdent leurs rapports privilégiés respectifs à l'addition et à la multiplication qui, du coup ne sont plus des lois de groupe.

#### élément symétrique

Notamment, il n'existe pas d'élément symétrique pour l'addition et la propriété  $x - x = 0$  ne tient plus pour  $x$  intervalle.

```
-Interval[{1, 2}]
Interval[{-2, -1}]
Interval[{1, 2}] - Interval[{1, 2}]
Interval[{-1, 1}]
```

Un phénomène similaire se produit avec la division.

```
Interval[{2, 5}] / Interval[{2, 5}]
Interval[{{2, 5}, {2, 5}}]
```

#### distributivité

La multiplication des intervalles n'est pas distributive par rapport à l'addition.

```
Interval[{-1, 2}] (Interval[{-3, 2}] + Interval[{4, 5}])

Interval[{-7, 14}]

(Interval[{-1, 2}] Interval[{-3, 2}]) + (Interval[{-1, 2}] Interval[{4, 5}])

Interval[{-11, 14}]
```

## Evaluation des expressions


A toute fonction numérique, définie par une expression symbolique - par exemple un polynôme - on peut associer une extension aux intervalles comme on l'a fait précédemment avec les opérateurs arithmétiques ou avec les fonctions usuelles.

Toutefois, deux expressions symboliquement équivalentes - et qui donc définissent la même fonction numérique produisent en général deux fonctions d'évaluation par intervalles différentes.

```
p[x_] := (x - 2) (x + 3);
Expand[p[x]]
HornerForm[p[x]]

-6 + x + x2

-6 + x (1 + x)
```



Remarquons que l'évaluation de  $p[x]$  par l'expression  $(x+1/2)^2 - 25/4$  est minimale car les bornes sont prises par des valeurs de la fonction. Un résultat classique (de l'analyse par intervalles) établit que c'est le cas si chacune des variables de l'expression qui prends des valeurs intervalles apparaît une fois seulement dans l'expression. Ce résultat assez intuitif est le seul dont on dispose.

Ce que l'on vient de voir nous conduit à essayer de trouver, parmi l'ensemble (infini) des expressions symboliques équivalentes associé à une fonction numérique, celle qui à la meilleure évaluation, c'est à dire aux intervalles résultats les plus étroits.

## Nombre d'occurrences des variables

Une première piste à explorer concerne le nombre d'occurrences de chaque variable. On considère l'exemple suivant

```
y = .
p1 = -6 x^3 + 4 x^3 z^3 + 15 x y^3 z^3 - 3 y^5 z^3 - 12 x^2 y z^3 - 3 x y z^3 + y^3 z^3;
b = {Interval[{-1., 2.}], Interval[{-2., 1.}], Interval[{-1.5, 1.5}]};
IEval[p1, {x, y, z}, b]

Interval[{-1681.5, 1639.5}]


NbOcc[p1, {x, y, z}]

{5, 5, 6}
```

On peut essayer de minimiser ces nombres d'occurrences - ou leur total.

```
p1s = OccSimplify[p1]
NbOcc[p1s, {x, y, z}]
IEval[p1s, {x, y, z}, b]
-6 x^3 + (4 x^3 - 3 x (1 + 4 x) y + (1 + 15 x) y^3 - 3 y^5) z^3
{5, 3, 1}
Interval[{-1222.5, 1180.5}]
```

Dans ce cas le résultat est intéressant, mais ce n'est pas systématique.



## Monotonie

Une autre piste consiste à évaluer les dérivées de l'expression pour analyser sa monotonie, afin de pouvoir appliquer le cas échéant la propriété suivante :

si  $f$  est croissante - resp. décroissante sur l'intervalle  $[a, b]$ ,  
alors  $f([a, b]) = [f(a), f(b)]$  - resp.  $f([a, b]) = [f(b), f(a)]$

Essayons avec l'exemple ci-dessous :

```
p2 = -6 x^3 - 4 x^3 z^3 + 15 x y^3 z^3 + 3 y^5 z^3 + 12 x^2 y z^3 + 3 x y z^3 + y^3 z^3;
b = {Interval[{0.5, 1}], Interval[{0.5, 1}], Interval[{0.5, 1}]}
IEval[p2, {x, y, z}, b]
{Interval[{0.5, 1}], Interval[{0.5, 1}], Interval[{0.5, 1}]}
Interval[{-9.57422, 33.1875}]
```

On dérive l'expression par rapport à chacune des variables concernées, et on évalue les dérivées.

```
IEval[D[p2, x], {x, y, z}, b]
IEval[D[p2, y], {x, y, z}, b]
IEval[D[p2, z], {x, y, z}, b]
Interval[{-28.8281, 37.125}]
Interval[{1.47656, 78.}]
Interval[{-9.44531, 101.625}]
```

Comme l'intervalle résultat de l'évaluation de la dérivée de l'expression par rapport à y est strictement positif, on peut conclure à la croissance du polynôme sur l'intervalle considéré. Ainsi le minimum (resp. le maximum) sera atteint pour la valeur minimale (resp. maximale) de y.

$$p2m = p2 / . y \rightarrow 0.5$$

$$p2M = p2 / . y \rightarrow 1.$$

$$-6 x^3 + 0.21875 z^3 + 3.375 x z^3 + 6. x^2 z^3 - 4 x^3 z^3$$

$$-6 x^3 + 4. z^3 + 18. x z^3 + 12. x^2 z^3 - 4 x^3 z^3$$

### borne inférieure

On considère maintenant le polynôme en deux variables qui va nous fournir le minimum,

$$p2ms = \text{OccSimplify}[p2m]$$

$$-6 x^3 - 4. (-1.94763 + x) (0.0754419 + x) (0.372193 + x) z^3$$

après simplification, ce polynôme s'avère croissant

$$D[p2ms, z] / . z \rightarrow b[[3]] / . x \rightarrow b[[1]]$$

$$\text{Interval}[\{1.42684, 25.6355\}]$$

on peut donc donner à z la valeur minimale

$$p2mm = p2ms / . z \rightarrow 0.5$$

$$-6 x^3 - 0.5 (-1.94763 + x) (0.0754419 + x) (0.372193 + x)$$

puis recommencer le même processus

$$D[p2mm, x] / . x \rightarrow b[[1]]$$

$$\text{Interval}[\{-18.0519, -2.97931\}]$$

et obtenir la borne minimale de la fonction de départ.

$$p2mmm = p2mm / . x \rightarrow 1.$$

$$-5.30078$$

### borne supérieure

C'est ensuite au tour du polynôme en deux variables qui va nous fournir le maximum,

$$p2Ms = \text{OccSimplify}[p2M]$$

$$-6 x^3 - 4. (-4.14411 + x) (0.2789 + x) (0.865208 + x) z^3$$

un processus similaire - à une simplification près - permet de venir à bout des calculs.

$$D[p2Ms, z] / . z \rightarrow b[[3]] / . x \rightarrow b[[1]]$$

$$p2MM = p2Ms / . z \rightarrow 1.$$

$$p2MMs = \text{OccSimplify}[p2MM]$$

$$D[p2MMs, x] / . x \rightarrow b[[1]]$$

$$p2MMM = p2MMs / . x \rightarrow 1.$$

$$\text{Interval}[\{10.03, 104.312\}]$$

$$-6 x^3 - 4. (-4.14411 + x) (0.2789 + x) (0.865208 + x)$$

$$-10. (-2.13219 + x) (0.293923 + x) (0.638266 + x)$$

$$\text{Interval}[\{0.678118, 38.8219\}]$$

$$24.$$

Le résultat final est un interval environ 50% moins large que celui obtenu avec l'évaluation "naturelle".



```
Interval[{p2mmm, p2MMM}]
Interval[{-5.30078, 24.}]
```

## Formules de Taylor

Lorsque l'on sait dériver, on peut aussi utiliser une version adaptée aux intervalles du développement de Taylor pour évaluer une fonction.

### premier ordre

$$f([X]) = f(m) + \left[ \frac{\partial f}{\partial x} \right]([X]) ([X] - m)$$

Dans la formule ci-dessus, le premier terme est l'évaluation de la fonction au point milieu de l'intervalle, le deuxième le produit de l'intervalle recentré par l'extension de sa dérivée aux intervalles évaluée sur l'intervalle tout entier (!).

Pour traiter un exemple à plusieurs variables, on itère le processus sur chacune des variables :

```
p1 /. x -> 0.75
D[p1, x]
D[p1, x] /. x -> b[[1]]
b[[1]] - 0.75
p1x = (p1 /. x -> 0.75) + (D[p1, x] /. x -> b[[1]]) (b[[1]] - 0.75)
-2.53125 + 1.6875 z3 - 9. y z3 + 12.25 y3 z3 - 3 y5 z3
-18 x2 + 12 x2 z3 - 3 y z3 - 24 x y z3 + 15 y3 z3
-3 y z3 + 15 y3 z3 + y z3 Interval[{-24, -12.}] + Interval[{-18, -4.5}] + z3 Interval[{3., 12}]
Interval[{-0.25, 0.25}]
-2.53125 + 1.6875 z3 - 9. y z3 + 12.25 y3 z3 - 3 y5 z3 +
Interval[{-0.25, 0.25}] (-3 y z3 + 15 y3 z3 + y z3 Interval[{-24, -12.}] + Interval[{-18, -4.5}] + z3 Interval[{3., 12}])
p1xy = (p1x /. y -> 0.75) + (D[p1x, y] /. y -> b[[2]]) (b[[2]] - 0.75)
-2.53125 - 0.606445 z3 +
Interval[{-0.25, 0.25}] (4.07813 z3 + z3 Interval[{-18., -9.}] + Interval[{-18, -4.5}] + z3 Interval[{3., 12}]) +
Interval[{-0.25, 0.25}] (-9. z3 + z3 Interval[{-15., -0.9375}] + z3 Interval[{9.1875, 36.75}]) +
Interval[{-0.25, 0.25}] (-3 z3 + z3 Interval[{-24, -12.}] + Interval[{0, 0}] + z3 Interval[{11.25, 45.}])
p1xyz = (p1xy /. z -> 0.75) + (D[p1xy, z] /. z -> b[[3]]) (b[[3]] - 0.75)
Interval[{-23.9144, 18.3402}]
```

C'est l'algorithme implémenté dans la fonction de bibliothèque **CenteredEval**.

```
NaturalEval[p1, {x, y, z}, b]
CenteredEval[p1, {x, y, z}, b]
Interval[{-23.8047, 18.957}]
Interval[{-23.9144, 18.3402}]
```

Notons que le gain n'est pas très important, mais qu'il s'améliore grandement avec le rétrécissement de l'intervalle de départ - un phénomène de convergence bien connu des amateurs de formules de Taylor de tous poils.

```
b1 = {Interval[{1.9, 2}], Interval[{0.9, 1}], Interval[{1.4, 1.5}]};
NaturalEval[p1, {x, y, z}, b1]
CenteredEval[p1, {x, y, z}, b1]
Interval[{-106.08, 45.5503}]
Interval[{-40.6544, -21.6928}]
```

```

b2 = {Interval[{1.9, 1.91}], Interval[{0.9, 0.91}], Interval[{1.49, 1.5}]};
NaturalEval[p1, {x, y, z}, b2]
CenteredEval[p1, {x, y, z}, b2]

Interval[{-38.2786, -23.4789}]

Interval[{-31.3969, -30.3772}]

```

## second ordre

$$[f]([X]) = f(m) + (X - m) \frac{\partial f}{\partial x}(x) + \frac{([X] - m)^2}{2} \left[ \frac{\partial^2 f}{\partial x \partial x} \right]([X])$$

On peut raffiner l'approximation de la fonction par une formule de Taylor en augmentant l'ordre de cette formule. La fonction de bibliothèque **TaylorEval** implémente ce calcul avec la même approche que précédemment - variable après variable.

```

TaylorEval[p1, {x, y, z}, b]
TaylorEval[p1, {x, y, z}, b1]
TaylorEval[p1, {x, y, z}, b2]

Interval[{-11.4277, 5.22482}]

Interval[{-37.6537, -24.0756}]

Interval[{-31.3713, -30.3974}]

```

L'amélioration des résultats en terme de largeur d'intervalles est plus sensible lorsque l'intervalle d'évaluation est grand, et s'atténue au fur et à mesure de sa diminution.

---

## Filtrage symbolique

L'évaluation est une étape cruciale qui intervient dans toutes les méthodes de l'analyse par intervalles. Mais des fonctionnalités de calcul symboliques - savoir manipuler, dériver des formules, faire des changements de variables - peuvent aussi aider à développer certaines de ces méthodes.

## Résolution de systèmes et filtrage

Un domaine de prédilection de l'analyse par intervalles est la résolution de systèmes d'équations. Il s'agit de déterminer, à l'intérieur d'un intervalle - resp. d'une boîte dans le cas d'un système à plusieurs variables - de départ, un ensemble d'intervalles (resp. de boîtes) suffisamment petit(e)s où le système peut-être satisfait.


La méthode d'analyse d'intervalles la plus populaire pour attaquer ce problème est une méthode itérative de recherche globale basée sur une boucle de bisection et d'évaluation éventuellement associée à un filtrage.

Un filtrage peut se définir comme l'opération suivante :

Etant donné une boîte  $b_i$  et un système à résoudre  $f(X) = 0$ ,  
trouver une boîte  $b_f$  incluse dans  $b_i$  telle que si il existe un élément  $x$  de  
 $b_i$  vérifiant  $f(x) = 0$ , alors nécessairement,  $x$  est un élément de  $b_f$ .

Il existe des filtres purement numériques, comme le filtrage de Newton. Prenons un exemple,

```
f[x_] := Sin[x] + x - 3
```



### Filtrage 2-B

Ce filtrage est par essence symbolique, même si souvent cet aspect est caché dans son implémentation qui s'appuie directement sur la représentation des systèmes d'équations.

L'idée est simple, il s'agit d'abord de décorréler les instances de chaque variables, en en introduisant de nouvelles pour obtenir des équations supplémentaires. Ensuite, on utilise ces équations supplémentaires pour filtrer.

Reprenons le même exemple,

$$f[x] == 0$$

$$-3 + x + \sin[x] == 0$$

On peut considérer la première instance de  $x$  comme indépendante de la deuxième, la remplacer par une autre variable,  $y$ .

$$\{\text{Solve}[\text{Replace}[f[x], x \rightarrow y, \{1\}] == 0, y][[1, 1]] /. \text{Rule} \rightarrow \text{Equal}, x == y\}$$

$$\{y = 3 - \sin[x], x = y\}$$

L'évaluation de la première équation donne une borne sur y en fonction de x, tandis que l'évaluation de la deuxième donne une borne - trivialement - sur x en fonction de y. Il suffit donc d'itérer.

```
Function[i, N[3 - Sin[x] /. x -> i]][Interval[{-10., 10.}]]
Interval[{2., 4.}]
```

Sur cet exemple, la convergence est longue.

```
NestList[Function[i, N[3 - Sin[x] /. x -> i]], Interval[{-10., 10.}], 25]
{Interval[{-10., 10.}], Interval[{2., 4.}], Interval[{2.0907, 3.7568}], Interval[{2.13213, 3.57713}],
Interval[{2.15346, 3.4219}], Interval[{2.165, 3.27665}], Interval[{2.1714, 3.13465}], Interval[{2.17501, 2.99305}],
Interval[{2.17705, 2.85201}], Interval[{2.17821, 2.71444}], Interval[{2.17887, 2.58572}], Interval[{2.17925, 2.47232}],
Interval[{2.17947, 2.37958}], Interval[{2.17959, 2.30962}], Interval[{2.17966, 2.26074}], Interval[{2.1797, 2.22872}],
Interval[{2.17973, 2.20874}], Interval[{2.17974, 2.19668}], Interval[{2.17975, 2.18955}],
Interval[{2.17975, 2.1854}], Interval[{2.17975, 2.183}], Interval[{2.17976, 2.18161}], Interval[{2.17976, 2.18082}],
Interval[{2.17976, 2.18037}], Interval[{2.17976, 2.18011}], Interval[{2.17976, 2.17996}]}
```

Remarquons qu'on aurait pu faire d'autre choix de variables, par exemple,

```
{Solve[Replace[f[x], Sin[x] -> z, {1}] == 0, x][[1, 1]] /. Rule -> Equal, z == Sin[x]}
{x == 3 - z, z == Sin[x]}
```

L'itération est un tout petit peu moins simple,

```
Function[i, N[3 - z /. z :> (Sin[x] /. x -> i)]][Interval[{-10., 10.}]]
Interval[{2., 4.}]
```

et sur cet exemple la convergence est identique ...

```
NestList[Function[i, N[3 - z /. z :> (Sin[x] /. x -> i)]], Interval[{-10., 10.}], 25]
{Interval[{-10., 10.}], Interval[{2., 4.}], Interval[{2.0907, 3.7568}], Interval[{2.13213, 3.57713}],
Interval[{2.15346, 3.4219}], Interval[{2.165, 3.27665}], Interval[{2.1714, 3.13465}], Interval[{2.17501, 2.99305}],
Interval[{2.17705, 2.85201}], Interval[{2.17821, 2.71444}], Interval[{2.17887, 2.58572}], Interval[{2.17925, 2.47232}],
Interval[{2.17947, 2.37958}], Interval[{2.17959, 2.30962}], Interval[{2.17966, 2.26074}], Interval[{2.1797, 2.22872}],
Interval[{2.17973, 2.20874}], Interval[{2.17974, 2.19668}], Interval[{2.17975, 2.18955}],
Interval[{2.17975, 2.1854}], Interval[{2.17975, 2.183}], Interval[{2.17976, 2.18161}], Interval[{2.17976, 2.18082}],
Interval[{2.17976, 2.18037}], Interval[{2.17976, 2.18011}], Interval[{2.17976, 2.17996}]}
```

... mais ce n'est pas toujours le cas.

## Sous-expressions


Un autre moyen symbolique de générer des équations redondantes supplémentaires qui vont permettre de faire du filtrage est de s'appuyer sur la représentation des expressions et des sous-expressions.

```
Eq = {x^2 + y + (y + x^2 + y^3 - 1)^3 - 2, ((x^2 + y^3) (x^2 + Cos[y]) + 14) / (x^2 + Cos[y]) - 8}
```

$$\left\{ -2 + x^2 + y + (-1 + x^2 + y + y^3)^3, -8 + \frac{14 + (x^2 + y^3)(x^2 + \cos[y])}{x^2 + \cos[y]} \right\}$$

Regardons la structure de la première équation.

```
Eq[[1]] // TreeForm
```



On peut introduire, comme le ferait un compilateur, des variables intermédiaires pour représenter les sous-expressions.

```
Experimental`OptimizeExpression[Eq, OptimizationSymbol -> X, OptimizationLevel -> 2]
```

```
Experimental`OptimizedExpression[
Block[{X$1183, X$1184, X$1185, X$1186, X$1187, X$1188, X$1190, X$1191, X$1192}, X$1183 = x^2; X$1184 = y^2;
X$1185 = y X$1184; X$1186 = -1 + X$1183 + y + X$1185; X$1187 = X$1186^2; X$1188 = X$1186 X$1187; X$1190 = Cos[y];
X$1191 = X$1183 + X$1190; X$1192 =  $\frac{1}{X$1191}$ ; {-2 + X$1183 + y + X$1188, -8 + X$1192 (14 + (X$1183 + X$1185) X$1191)}]]
```

L'introduction de certaines de ces nouvelles variables fournit de nouvelles équations qui peuvent rendre le filtrage du système plus efficace. Le choix de ces variables et la possibilité de prévoir le gain induit est un problème de recherche en cours.

## Méthodes algébriques

Dans le cas particulier de systèmes algébriques, on peut aussi faire appel à des méthodes algébriques complexes telles que les *bases de Gröbner* qui enrichissent les systèmes polynomiaux par des équations supplémentaires particulièrement adaptées à leur résolution.

```
GroebnerBasis[{x^25 + 2 x y + 1, x y - x}, {x, y}]
```

```
{-1 + y, 1 + 2 x + x^25}
```

```
GroebnerBasis[{x^2 + y^2 + z^2 - 1, x y - z + 2, z^2 - 2 x + 3 y}, {x, y, z}]
```

```
{1024 - 832 z - 215 z^2 + 156 z^3 - 25 z^4 + 24 z^5 + 13 z^6 + z^8, -11552 + 2560 y + 2197 z + 2764 z^2 + 443 z^3 + 728 z^4 + 169 z^5 + 32 z^6 + 13 z^7,
-34656 + 5120 x + 6591 z + 5732 z^2 + 1329 z^3 + 2184 z^4 + 507 z^5 + 96 z^6 + 39 z^7}
```

```
GroebnerBasis[{x^2 + y^2 + z^2 - 1, x y - z + 2, z^2 - 3 + x, x - y^2 + 1}, {x, y, z}]
```

```
{1}
```

```
GroebnerBasis[{x^2 + y^2 + z^2 - 1, x y - z + 2}, {x, y, z}]
```

```
{4 - y^2 + y^4 - 4 z + z^2 + y^2 z^2, -2 x - y + y^3 + x z + y z^2, 2 + x y - z, -1 + x^2 + y^2 + z^2}
```

Une description, même intuitive de la théorie des bases de Gröbner dépasse le cadre de ce cours.

## Exécution symbolique des algorithmes

C'est à la fois l'utilisation du calcul symbolique qui semble la plus prometteuse pour améliorer les algorithmes d'analyse par intervalles ... et celle qui est la moins connue - pour l'instant.

La plupart des algorithmes mettent en oeuvre des calculs - notamment des calculs matriciels - qui sont actuellement menés numériquement soit avec des nombres soit avec des intervalles.

Par exécution symbolique des algorithmes, on entend mener - tant que faire se peut - les calculs de façon symbolique, pour retarder les évaluations par intervalles et ainsi améliorer leur qualité.

Illustrons cette idée sur un exemple simple.


### Filtrage de Newton en dimension 2

On considère le problème suivant,

```
g1[{x_, y_}] := x^2 + y^2 - 1
g2[{x_, y_}] := (x - 1)^2 + y^2 - 1
b = {Interval[{0.45, 0.65}], Interval[{0.75, 0.95]}};
```

qui consiste à localiser le point d'intersection ci-dessous.

```
ContourPlot[{g1[{x, y}] == 0, g2[{x, y}] == 0}, {x, 0.15, 0.95}, {y, 0.15, 0.95}]
```



Pour résoudre ce système, on va utiliser un filtrage de Newton bi-dimensionnel, en appliquant la formule suivante

$$b_{n+1} = \bigcap (b_n, \text{milieu}(b_n) - J(b_n)^{-1} g(\text{milieu}(b_n)))$$

où  $g$  représente le système et  $J$  sa jacobienne.

```
(J = D[{g1[{x, y}], g2[{x, y}]}, {{x, y}}]) // MatrixForm
```

$$\begin{pmatrix} 2x & 2y \\ 2(-1+x) & 2y \end{pmatrix}$$

Un essai classique fournit un résultat catastrophique.

```
m = MidPoint[b]
J /. {x -> b[[1]], y -> b[[2]]} // MatrixForm
Inverse[J /. {x -> b[[1]], y -> b[[2]]}] // MatrixForm
{g1[b], g2[b]}
m - Inverse[J /. {x -> b[[1]], y -> b[[2]]}] . {g1[b], g2[b]}
{0.55, 0.85}

( Interval[{0.9, 1.3}] Interval[{1.5, 1.9}] )
( Interval[{-1.1, -0.7}] Interval[{1.5, 1.9}] )

( Interval[{0.328947, 0.791667}] Interval[{-0.791667, -0.328947}] )
( Interval[{0.153509, 0.458333}] Interval[{0.197368, 0.541667}] )

{Interval[{-0.235, 0.325}], Interval[{-0.315, 0.205}]}

{Interval[{0.0433333, 0.898333}], Interval[{0.59, 1.12833}]}
```

Si l'on mène les calculs symboliques plus loin, et notamment dans le calcul de l'inverse de la jacobienne - bien regarder les détails du code ci-dessus et ci-dessous - on obtient de meilleurs résultats qui ne sont, malheureusement sur cet exemple, pas suffisamment bons pour assurer la convergence.

```
J /. {x -> b[[1]], y -> b[[2]]} // MatrixForm
Inverse[J /. {x -> b[[1]], y -> b[[2]]}] // MatrixForm
{g1[b], g2[b]}
m - (Inverse[J /. {x -> b[[1]], y -> b[[2]]}] . {g1[b], g2[b]})

( Interval[{0.9, 1.3}] Interval[{1.5, 1.9}] )
( Interval[{-1.1, -0.7}] Interval[{1.5, 1.9}] )

( Interval[{0.184211, 0.366667}] Interval[{-0.236842, 0.433333}] )

{Interval[{-0.235, 0.325}], Interval[{-0.315, 0.205}]}

{Interval[{0.23, 0.77}], Interval[{0.642, 1.07267}]}
```